KFKI-1979-60

V.FAJER
L.ALVAREZ

ORGANIC SCINTILLATOR EFFICIENCY USING A MONTE CARLO CODE

Hungarian Academy of Sciences

CENTRAL RESEARCH INSTITUTE FOR PHYSICS

BUDAPEST

KFKI-1979-60

ORGANIC SCINTILLATOR EFFICIENCY USING A MONTE CARLO CODE

V. Fajer

Roland Eötvös University, Budapest
Budapest, VIII. Puskin u. 5-7.

L. Alvarez

Central Research Institute for Physics
Hungarian Academy of Sciences,
Budapest, XII. Konkoly Thege ut 29-33.

On leave from Nuclear Research Institute,
Academy of Sciences of Cuba, Managua, Havana,
P.O.B. 6122, Cuba

HU ISSN 0368 5330
ISBN 963 371 582 2

ABSTRACT

The TAYRA code was developed for calculating the efficiency organic scintillators for neutrons in the energy range 0.08 to 15 MeV. The algorithm uses the Monte Carlo method and considers, for the simulation, elastic scattering on hydrogen and carbon, inelastic scattering on carbon, and the reactions: $^{12}C(n,n')^{12}C^* \rightarrow 3\alpha$ and $^{12}C(n,\alpha)^{9}Be$.

The code is written in FORTRAN IV for an ES 1040 computer. The results obtained using TAYRA are compared with experimental and calculated efficiency data.

АННОТАЦИЯ

Программа ТАУРА разработана для определения эффективности органических нейтронных детекторов в диапазоне 0.08-15 Мэв. В программе использовался метод Монте-Карло. В испытании учитываются упругие и неупругие рассеяния на H и C и тоже реакция $^{12}C(n,n')^{12}C^* \rightarrow 3d$ и $^{12}C(n,\alpha)$ Be^9.

Программа написана на языке ФОРТРАН-IV, а использована на машине ЕС-1040. Результаты полученные программой ТАУРА были сопоставлены с экспериментальными и расчетными данными других авторов.

KIVONAT

A TAYRA nevü számitógépi program szerves szcintillátorok hatásfokát számitja, E_n = 0.08 – 15 MeV energiáju neutronok esetében. A program a Monte Carlo módszert használja fel a neutronok hidrogénen és szénen történõ rugalmas, ill. rugalmatlan szórásának és a $^{12}C(n,n')^{12}C^* \rightarrow 3\alpha$ és a $^{12}C(n,\alpha)^{9}Be$ reakcióknak a szimulálására.

A program nyelve FORTRAN IV, ES 1040 tipusu számitógépre alkalmazva.

A program által számitott hatásfok értékeket kisérleti eredményekkel hasonlitottuk össze.

INTRODUCTION

Organic scintillators have many applications in experimental cross section and nuclear reaction studies. Precise knowledge of the efficiencies of these scintillators is necessary in order to obtain the minimum possible errors in the final results.

The efficiency value is, at present, obtained in two ways: experimental and calculated.

The experimental way frequently uses the time-of-flight technique; the calculated way offers two possibilities, the employment of semi-empirical formulae and the Monte Carlo method. Inherent in the first possibility is the problem that the required correction factors are approximations and they are valid only in certain energy ranges. This means that larger errors arise than those desirable for accurate calculations of the efficiency.

The Monte Carlo method, which has been widely employed for this purpose, is the best way to calculate the efficiency value. It is based on the simulation of the physical events which take place inside the scintillator because of an incident neutron.

In order to improve the physical model, TAYRA code employs the cross sections obtained in small energy intervals. From 0.1 to 15 MeV a constant energy interval of 0.1 MeV was selected, and from 10 to 90 keV a constant interval of 10 keV was taken.

The lower energy range permits one to extend the efficiency calculation to a zone where the neutron detection has become feasible by fast plastic scintillators because of the development of low noise fast photomultipliers.

2. NEUTRON HISTORY ANALYSIS

TAYRA can calculate the efficiency for two geometrical arrangements for cylindrical scintillators: for a parallel beam of monoenergetic neutrons incident on the lateral surface of the scintillator, and for a beam of neutrons incident on the circular flat surface of the cylinder (*Fig.1*).

In order to begin the neutron history, it is necessary to determine the starting coordinates (x,y,z). In the case of perpendicular incidence to the cylinder axis these are:

$$x = A\sqrt{1 - R1^2}\,,$$
$$y = A\ R1\,, \qquad (1)$$
$$z = B\ R2\,,$$

where A is the scintillator radius, B is the half-height and R1 and R2 are two random numbers obtained from a uniform distribution between 0 and 1.

The starting coordinates are points on the scintillator surface. In this case the direction cosines of the initial incidence direction are: $\cos\alpha = -1$, $\cos\beta = 0$, and $\cos\gamma = 0$.

With parallel incidence to the cylinder axis, the (x,y) coordinates are calculated in the same way and the z coordinate is:

$$z = B \qquad (2)$$

and the direction cosines are: $\cos\alpha = 0$, $\cos\beta = 0$ and $\cos\gamma = -1$.

In order to determine the next neutron coordinates, it is necessary to know the neutron free path in the scintillator, which is obtained by:

$$\lambda(E) = \frac{1}{\Sigma_T(E)}\,, \qquad (3)$$

where $\Sigma_T(E)$ is the total macroscopic cross section. With this value it is possible to determine the neutron path length, $\rho(E)$, by:

$$\rho(E) = -\lambda(E)\ \ln R3\,, \qquad (4)$$

where R3 is another random number from the same distribution between 0 and 1. This value, $\rho(E)$, permits one to decide wheter the neutron escapes from the scintillator or, alternatively one can determine the new coordinates which correspond to the point where the first interaction takes place.

Since the scintillator can be considered as a mixture of hydrogen and carbon atoms, the most probable interactions of fast neutrons with these nuclei to be taken into account in the energy range 0.02 to 15 MeV, are:

a. Elastic scattering on hydrogen nuclei,
b. Elastic scattering on carbon nuclei,
c. Inelastic scattering on carbon nuclei,
d. The reaction $^{12}C\ n,n'\ C^{*} \rightarrow 3\alpha$,
e. The reaction $^{12}C(n,\alpha)Be^{9}$.

The cross section values were obtained from reference [1].

2.1 Elastic scattering on hydrogen nuclei

Elastic scattering on hydrogen nuclei is the most probable interaction below 10 MeV. From this collision are obtained a scattered neutron and a recoil proton, and their energies after the scattering are:

$$E_P = \frac{1}{2} E_o(1 - \cos\kappa)$$
$$E_n = \frac{1}{2} E_o(1 + \cos\kappa) \qquad (5)$$

where E_o is the incident neutron energy and κ is the scattering angle in the centre of mass system, where the elastic scattering is isotropic [2].

The magnitude of the light output produced by the recoil proton is one of the most problematic aspects of the physical model. In order to improve this aspect, light output values recently obtained or confirmed, have been employed. Tabulated values obtained from [3] have been used in the energy range 20 to 200 keV, and the following semi-empirical formula given in [4] is employed in the range 0.2 to 15 MeV:

$$L_P = a_1 \left[1 - \exp\left(- a_2 E_P^{a_3}\right)\right] + a_4 E_P, \qquad (6)$$

where a_1, a_2, a_3 and a_4 are parameters which depend on the scintillator type. This physical event is considered in the TAYRA code in the subroutine HYDR.

2.2 Elastic scattering on carbon nuclei

The result of this collision is a recoil carbon nucleus and a scattered neutron with energies:

$$E_c = 0.142\ E_o\ (1 - \cos\kappa), \tag{7}$$

$$E_n = E_o - E_c \tag{8}$$

In the scintillator the carbon nucleus produces a light output whose magnitude is calculated, as is shown in [5], by:

$$L_c = 0.01\ E_c \tag{9}$$

The anisotropy of this event in the centre of mass system is taken into account. The angular distribution values were taken from [1]. The probabilities of the different directions are calculated from these angular distributions. This event is simulated in the TAYRA code in the subroutine ELAC.

2.3 Inelastic scattering on carbon nuclei

This interaction produces a gamma ray of 4.43 MeV and an outgoing neutron, and it becomes important for incident neutrons of low energies. The light output produced by the gamma ray is not taken into account because the detection efficiency is poor for its energy, for details see ref. [5].

The main effect of this interaction is to decrease the neutron energy in 4.43 MeV without changing its direction.

2.4 The reaction $^{12}C(n,n')C^* \rightarrow 3d$

This reaction begins to be important above 11 MeV and remains with low probability at 14 MeV. Nevertheless, it gives three alpha particles with energies, in the laboratory system, in the range 1 to 3 MeV; each alpha particle produces a light output of approximately 100 keV in electron equivalent energy.

In the TAYRA code, when this event takes place, the produced neutron is detected if the selected threshold is lower than 3 MeV.

2.5 The reaction $^{12}C(n,\alpha)Be^9$

In this reaction an alpha particle and a beryllium nucleus are produced, whose energies are:

$$E_\alpha = (E_n - Q)(B + D + 2\sqrt{A\,C}\,\cos\kappa)$$
$$E_{Be} = (E_n - Q)(A + C - 2\sqrt{A\,C}\,\cos\kappa), \qquad (10)$$

where A,B,C and D are coefficients which depend on the mass of the nuclei and particles that take part in the reaction, and they also depend on the neutron energy, E, and the Q-value of the reaction, (-5.71 MeV). These coefficients are given in ref.[6].

The light output produced by the ^{9}Be nucleus can also be obtained using the formula (9). The light output of the alpha particle can be determined by the following formula used in ref.[2]:

$$L_\alpha = 0.046\,E_\alpha + 0.007\,E_\alpha^2 \qquad (11)$$

The anisotropy in the centre of mass system is also considered in this event; a constant angular distribution, measured and fitted in ref. [7], was used for the whole energy range. These calculations are taken into account in the TAYRA code in the subroutine ALPHAN.

The event which takes place is determined thereby generating a random number. This number is then compared with the occurrence probability of each event. The probabilities are previously assigned taking into account the cross sections.

The new direction of the scattered neutron, after each interaction, is calculated to enable the new coordinates to be located.

When the interaction is considered isotropic in the centre of mass system, the cosine of polar scattering angle of the outgoing particle, $\cos\kappa$, and the azimuthal angle ϕ can be randomly generated, $\cos\kappa$ is generated from a uniform distribution between -1 and +1 by:

$$\cos\kappa = 2\ R5 - 1, \tag{12}$$

where R5 is a random number between 0 and 1. The cosine of ϕ only takes values between 0 and 1.

If the interaction is considered anisotropic, $\cos\kappa$ is taken from the correspondent angular distribution.

The scattering angle in the laboratory system, θ, is calculated by the following formulae:

$$\cos\theta = (1 + m_2\cos\kappa)/\sqrt{1 + 2\ m_2\cos\kappa + m_2^2}\ , \tag{13}$$

$$\sin\theta = m_2\sin\kappa/\sqrt{1 + m_2\cos\kappa + m_2^2}\ ,$$

where m_2 is the mass of the scatterer nucleus.

The direction cosines after the scattering are, as is shown in ref. [8]:

$$\cos\alpha' = \cos\alpha\cos\theta + (\cos\gamma\cos\alpha\sin\theta\cos\phi - \cos\beta)\sin\theta\sin\phi/(1-\cos^2\gamma)^{1/2},$$

$$\cos\beta' = \cos\beta\cos\theta + (\cos\gamma\cos\beta\sin\theta\cos\phi + \cos\alpha\sin\theta\sin\phi)/(1-\cos^2\gamma)^{1/2},$$

$$\cos\gamma' = \cos\gamma\cos\theta - (1 - \cos^2\gamma)^{1/2}\ \sin\theta\cos\phi, \tag{14}$$

except when $(1-\cos^2\gamma)$ approaches zero, in which case the following equations are used:

$$\cos\alpha' = \sin\theta\ \cos\phi$$

$$\cos\beta' = \sin\theta\ \sin\phi \tag{15}$$

$$\cos\gamma' = \cos\gamma\ \cos\phi$$

These equations are employed in the subroutine COSINE.

The history is interrupted when the neutron escapes from the scintillator or its energy is diminished below the selected cut-off (20 keV), then the neutron is absorbed by the scintillator. When a light pulse or a sequence of pulses, whose total magnitude is greater than the light threshold, is produced, the history of the neutron is also interrupted.

Finally the efficiency value is obtained from the ratio of the detected neutrons to the number of histories followed, and the relative and absolute errors are also calculated by the TAYRA code.

3. RESULTS AND CONCLUSIONS

The Monte Carlo program was employed to calculate efficiency values of different organic scintillators. The results were compared with some available experimental data from refs.[9] and [10]; the agreement between experimental and Monte Carlo values was found to be reasonable as is shown in *Tab. 1.* and *Fig.2.*

The experimental data taken from ref.[9] were also compared with the Monte Carlo predictions of [5] and it was found that the TAYRA code gives also good agreement with these measurements.

In general, one can observe that the disagreement between the measured and the calculated efficiencies is around 10% overall, and this seems to be produced because of the uncertainties in the inelastic n-C cross sections [11], and in the light output values of the α particles and the ^{12}C and ^{9}Be nuclei. Furthermore, the Monte Carlo calculations have not taken into account the light attenuation effects inside the scintillator and in the optical coupling.

It was obviously important to be extremely careful in the simulation of each experimental condition of the measurements, especially in the selection of the light threshold.

The results of TAYRA were also compared with some efficiency values obtained by other Monte Carlo calculations, from refs. [3], [6] and [12]. The comparison with the predictions of [3] was possible because of the use of a small energy interval, 20 keV. in the range 20 to 200 keV, and also because the cut-off energy was lowered, (*Fig.3a*).

In this way the TAYRA code can calculate the efficiency values in the energy range from 80 to 500 keV, which is very important nowadays because the development of low noise photomultiplier tubes and organic scintillators with large light output has permitted the use of these neutron detectors in this range.

The results given in [3] were obtained using a modified version of the O5S code developed by Textor and Verbinski, ref.[6]. The results from TAYRA were also compared with the predictions of the O5S code, as is shown in *Tab.2* and *Fig.3b*.

The very small discrepancies between the TAYRA results and the other predictions seem to be because of the employment of different cross section values for the n-C interactions and also because of the introduction in the TAYRA code of a semiempirical formula for the light output of the recoil proton, formula (6) given in ref.[4]. This formula gives good agreement with the experimental data of other authors.

Finally, *Fig.4.* shows the efficiency curve for an NE-102 A scintillator obtained using an interval of 0.5 MeV in the energy range 1 to 15 MeV. These results were fitted by:

$$E_F = (1 - T/E_n)(A_o + A_1E_n + A_2E_n^2 + A_3E_n^3 + A_4E_n^4) \qquad (16)$$

where E_F is the efficiency, T is the threshold in proton energy and the values of the parameters were: $A_o = 73.38$, $A_1 = -20.75$, $A_2 = 3.32$, $A_3 = -0.24$ and $A_4 = 0.64$.

The execution time for a run was approximately 30 sec for 10 000 histories and the relative error was less than 3% in all the runs.

ACKNOWLEDGEMENTS

We are grateful to A. Kiss and F. Deák for their comments and helpful discussions related to this work. The help of

Lic. J.M. Hernandez and B. Barrios in the preparation of the code is appreciated. We wish to thank Dr. I.Fodor and Dr.J.Sziklai for their kind hospitality in the Central Research Institute for Physics, Budapest, where it was possible to develop the TAYRA code.

APPENDIX

Summary of program components and input data

System: ES-1040

Language: FORTRAN IV

Components:

1. Main program: TAYRA
2. Subroutines:
   a. CARG. Provides the cross sections for the considered events and also the necessary angular distributions and the tabulated data of light output in the energetic range from 20 to 200 keV. This subroutine also stores the scintillator data.
   b. PROBA(E). Chooses the probabilities for the neutron energy and these are stored in the blank COMMON.
   c. COSENO(SITA,ALF,BET,GAM). Computes the centre of mass cosines of the polar scattering direction in both cases: isotropic and anisotropic scattering,and uses them to calculate the new direction cosines in the laboratory system.
   d. HYDR(ENE,SITA,P). Computes the energies of the scattered neutron and the recoil proton when elastic scattering on hydrogen takes place, and also computes the light output due to the recoil proton.
   e. ELAC(ENE,SITA,P). Computes the same as HYDR in the case of elastic scattering on carbon.
   f. ALPHAN(P). Calculates the energies and the light output of the alpha particle and the beryllium nucleus when the reaction $^{12}C(n,\alpha)Be^{9}$ takes place.
3. FORTRAN functions:
   a. RANDOM(KS). Computes random numbers uniformly distributed between 0 and 1, using any odd number as input parameter.

b. SEGN(X). Attributes values +1 or -1 if the argument X is negative or positive.

c. ALAMDA(ENE). Computes the mean free path of the neutron depending on its energy ENE.

4. INPUT DATA:

a. CARD SET 1, one card, OL(10)
FORMAT(10F7.0)
OL(I): light output in the energetic range from 20 to 200 Kev with 20 Kev as step.

b. CARD SET 2, two cards, SEM(8,2)
FORMAT(8F10.0)
SEM(J,I): cross sections in the energetic range from 20 to 90 Kev with 10 Kev as step, J is the index for the energy and I is for the interaction channels in this energetic range (elastic scattering on hydrogen and carbon).

c. CARD SET 3, 75 cards, SE(150,5)
FORMAT(10F8.0)
SE(J,I): cross sections in the energetic range from 100 Kev to 15 Mev with 0.1 Mev as interval, J is the index for the energy and I is for the interaction channels.

d. CARD SET 4, 3 cards, PE(42)
FORMAT(16F5.0)
PE(I): energetic groups for the angular distributions of the elastic scattering on carbon.

e. CARD SET 5, 126 cards, AD(21,42)
FORMAT(7F10.0)
AD(J,I): angular distributions of the elastic scattering on carbon, J is the index for the energy and I for the value of the cosine of the scattering angle in the centre of mass system.

f. CARD SET 6, 3 cards, ADA(21)
FORMAT(7F10.0)
ADA(I): angular distributions for 14 Mev of the reaction $^{12}C(n,\alpha)Be^{9}$ which is taken to remain the same in the whole energy range.

g. CARD SET 7, one card, HN,RHC,A,C,G

FORMAT(5F10.0)

HN: number of hydrogen atoms/cm-barn,

RHC: hydrogen/carbon ratio,

A: scintillator radius,

B: scintillator half height,

G: geometry code, G=1 if the neutron incidence is perpendicular to the cylinder axis and G=2 if the neutron incidence is parallel to this one.

CARD SET 8, one card, RN

FORMAT(F10.0)

RN: number of neutron histories.

i. CARD SET 9, one card, EMI,UMBRAE

FORMAT(2F10.0)

EMI: cut off energy,

UMBRAE: energetic threshold.

j. CARD SET 10, as many cards as you need, EI

FORMAT(F10.0)

EI: initial neutron energy, if it is necessary to consider more than one energetic threshold one must put a last card of the set 10 with EI=0, and after a card of the set 9 and so on. If one needs to finish the sequence it is necessary to put a last card of the set 10 with EI = -1.

NOTE: When using the TAYRA code to calculate different efficiencies for different experimental arrangements, the only change is to alter card sets 7,8,9 and 10.

REFERENCES

1) P. Vértes, FEDGROUP-A program system for producing group constants from evaluated data files disseminated by IAEA: INDC(HUN)-13/L+sp, 1976.

2) R. Batchelor and W.B. Gilboy, Nucl. Instr. and Meth. 13(1961)70.

3) C. Renner, N.W. Hill, G.L. Morgan, K. Rush and J.A. Harvey, Nucl. Instr. and Meth. 154 (1978) 525.

4) R. Madey, F.M. Waterman, A.R. Baldwin, J.N. Knudson, J.D. Carlson and J. Rapaport, Nucl. Instr. and Meth.151(1978) 445.

5) R.J. Schuttler, ORNL-3888, (1966).

6) R.E. Textor and V.V. Verbinski, ORNL-4160, (1968).

7) M.L. Chaterjee and B. Sen, Nucl. Phys. 51 (1964) 583.

8) G.B. Beam, L. Wielopolski, R.P. Gardner and K. Verghese, Nucl. Instr. and Meth. 154 (1978) 501.

9) T.A. Love, R.T. Santoro, R.W. Peelle and N.W. Hill, Rev. Sci. Instr. 39 (1968) 541, No.4.

10) P. Leleux, P.C. Macq. J.P. Meulders and C. Pirart, Nucl. Instr. and Meth. 116 (1974) 41.

11) A. Del Guerra, Nucl. Instr. and Meth. 135 (1976) 337.

12) V.V. Verbinski, W.R. Burrus, T.A. Love, W. Zobel, N.W. Hill and R. Textor, Nucl. Instr. and Meth. 65 (1968) 8.

Table 1.

Comparison of calculated and experimental efficiency values for a bias threshold – 180 keV p.e.q. for two NE-213 scintillators and neutron incidence parallel to the cylinder axis.

Scintillator dimensions cm	Neutron energy MeV	TAYRA code %	Experimental Ref.[9] %	Calculated Ref.[5] %
12 x 2.61	2.7	16.97 + 0.20	17.22 + 0.11	---
	14.5	10.06 + 0.16	10.62 + 0.08	9.70
12 x 6.10	2.7	30.40 + 0.28	33.75 + 0.15	---
	14.5	20.89 + 0.23	22.60 + 0.20	21.70

Table 2.

Comparison of calculated efficiencies by two different codes for an NE-213 scintillator, bias threshold = 0 keV and neutron incidence perpendicular to the cylinder axis.

Scintillator dimensions cm	Neutron energy MeV	TAYRA code %	05S code, Ref. [6] %
5.419 x 9.174	2.39	48.12 + 0.35	49.1



Fig.1. Geometrical arrangements that can be simulated by the TAYRA code.



Fig.2. Comparison with experimental data. Parallel neutron incidence to the cylinder axis. Diameter: 20 cm. Height: 2 cm.



Fig.3. Comparison of TAYRA calculations with other predictions: (a) O5 modified; (b) O5S code. Parallel neutron incidence to the cylinder axis.



Fig.4. Efficiency curve fitted by expression (13). Perpendicular neutron incidence to the cylinder axis. Diameter: 5 cm. Height: 10 cm.

T A Y R A   L I S T I N G

MAIN DATE = 79150 11/16/27

```
C-----MONTE CARLO CODE FOR CALCULATING EFFICIENCIES OF ORGANIC SCINT.
C
      COMMON/LIGHT/OL(10)
      COMMON/ANI/ENE,PE(42),AD(21,42),ADN(21)
      COMMON/JM/HN,RHC,A,B,G
      COMMON/SEMI/KS
      COMMON/HILDA/RM1,RM2
C-----IN THE BLANK COMMON ARE ACUMULATED THE PROBABILITIES FOR
C     THE EVENTS CONSIDERED
      COMMON PNP,PEC,PNG,P3A,PA
      RM1=1.
      CALL CARG
      READ(5,900)RN
  400 READ(5,900)EMI,UMBRAE
      WRITE(6,901)G
      WRITE(6,800)A,B
      WRITE(6,802)RN
      WRITE(6,803)EMI,UMBRAE
  405 READ(5,900)EI
      KS=157
      WRITE(6,805)EI
      IF(EI)410,400,420
  410 STOP
  420 RI=0
      RND=0
      RH=0
      RC=0
      RG=0
      RE=0
      XL=0
C-----IF THE ENERGETIC THRESHOLD IS LARGER THAN 0.2 MEV, THE LIGHT
C     THRESHOLD IS CALCULATED BY A SEMI-EMPYRICAL FORMULA, IF IT IS
C     LOWER THAN 0.2 MEV TABULATED DATA ARE TAKEN.
      IF(UMBRAE)500,500,501
  500 UMBRAL=0.00001
      GO TO 140
  501 IF(UMBRAE-0.2)502,502,503
  502 UMBRAL=OL(IFIX(((UMBRAE+.01)/2.)*100.))
      GO TO 140
  503 T1U=.95*UMBRAE
      T2U=-.1*EXP(.9*ALOG(UMBRAE))
      T3U=-8.*(1.-EXP(T2U))
      UMBRAL=T1U+T3U
  140 ENE=EI
C-----THE POSITION OF THE INCIDENT NEUTRON IS CALCULATED USING THE
C     RANDOM NUMBERS R1 AND R2. IF THE KEY G=1 THE INCIDENCE IS PERP. TO
C     THE CYLINDER AXIS, IF G=2 IS //, AND THE PROCEDURE IS DIFERENT.
C-----THE DIRECTION COSINES ARE (-1,0,0) AND (0,0,-1) RESPECTIVELY.
      R1=RANDOM(KS)
      X=A*SQRT(1.-R1**2)
      Y=A*R1
      IF(G-2.)1101,1102,1102
 1101 ALF=-1
      BET=0
      GAM=0
      R2=RANDOM(KS)
      Z=B*R2
      H=2.*X
```

```
      GO TO 1103
 1102 ALF=0
      BET=0
      GAM=-1
      Z=B
      H=2.*B
C-----THE TRAJECTORY RO IS COMPUTED TO KNOW IF THE NEUTRON ESCAPES
C     OR NOT.
 1103 R3=RANDOM(KS)
      RO=-ALAMDA(ENE)*ALOG(R3)
      IF(RO-H)1106,1100,1100
 1106 IF(G-2.)1104,1105,1105
 1104 X=X+RO
      GO TO 190
 1105 Z=Z+RO
      GO TO 190
 1100 RF=RF+1.
      GO TO 370
C     THE RANDOM NUMBER R4 SERVES TO DECIDE WHICH EVENT TAKES PLACE.
  190 R4=RANDOM(KS)
      IF(R4-PNP)230,230,220
  220 IF(R4-(PNP+PEC))240,240,222
  222 IF(R4-(PNP+PEC+PNG))223,223,225
  225 IF(R4-(PNP+PEC+PNG+P3A))362,362,226
C     REACCION C12(N,ALFA)BE9
  226 CALL ALPHAN(P)
      GO TO 270
C     INTERACCION INELASTICA CON EL CARBONO
  223 ENE=ENE-4.4
      IF(ENE-EMI)370,224,224
  224 CALL PROBA(ENE)
      RG=RG+1.
      GO TO 190
C      INTERACCION NP
  230 RM2=1
C-----THE SUBROUTINE COSENO COMPUTES THE NEW DIR. COSINES OF THE SCA-
C     TTERED NEUTRON ALF, BET AND GAM.
      CALL COSENO(SITA,ALF,BET,GAM)
      CALL HIDR(ENE,SITA,P)
      RH=RH+1
      GO TO 270
C     INTERACCION ELASTICA DEL CARBONO
  240 RM2=12
      CALL COSENO(SITA,ALF,BET,GAM)
      CALL ELAC(ENE,SITA,P)
      RC=RC+1.
C-----IF THE LIGHT PULSE IS > THRESHOLD,THE NEUTRON IS COUNTED, IF IT IS
C     NOT THIS PULSE IS ACCUMULATED IN XL IN ORDER TO BE ADDED WITH OTHER
C     PULSES PRODUCED FOR THE SAME NEUTRON DURING ITS "HISTORY"
  270 IF(P-UMBRAL)120,360,360
  120 XL=XL+P
C-----THE HISTORY OF THE NEUTRON IS FOLLOWED UNTIL IT IS ABSORBED
C     ESCAPED, OR ITS ENERGY IS DIMINUISHED TO 20 KEV
      IF(ENE-EMI)370,290,290
  290 R10=RANDOM(KS)
C-----THE NEW POSITION (X,Y,Z) OF THE NEUTRON IS COMPARED WITH THE
C-----SCINTILLATOR DIMENSIONS
      RO=-ALAMDA(ENE)*ALOG(R10)
```

```
      X=X+(RO*ALF)
      Y=Y+(RO*BET)
      Z=Z+(RO*GAM)
      IF(ABS(Z)-B)310,310,370
  310 RAD=SQRT(X**2+Y**2)
      IF(RAD-A)340,340,370
  340 IF(XL-UMBRAL)190,360,360
C     REACCION C12(N,3ALPHA)
C     THIS REACTION ES ONLY TAKEN INTO ACCOUNT IF THE THRESHOLD IS
C     IS GRATER THAN 0.22
  362 IF(UMBRAL-0.22)370,360,360
  360 XL=0
      RND=RND+1.
C----IN RI ARE ACUMULATED THE FOLLOWED NEUTRONS, AND IN RND WHICH OF
C        THEM WERE DETECTED. RN IS NUMBER OF SELECTED HISTIRIES
  370 RI=RI+1.
      IF(RI-RN)140,390,390
  390 EFIC=RND/RN*100.+.0001
      ER=100.*SQRT(RND)/RND+.00001
      RC=RC/RN*100.+.0001
      RH=RH/RN*100.+.0001
      RG=RG/RN*100.+.0001
      RE=RE/RN*100.+.0001
      WRITE(6,806)RH
      WRITE(6,807)RC
      WRITE(6,808)RG
      WRITE(6,809)RE
      WRITE(6,811)EFIC
      WRITE(6,812)ER
      WRITE(6,902)
  800 FORMAT(5X,'SCINTILLATOR RADIUS(CM)',F7.3,10X,'HALF-HEIGHT(CM)',F7.
     13//)
  802 FORMAT(5X,'NUMBER OF HISTORIES',F8.0//)
  803 FORMAT(5X,'CUT-OFF ENERGY(MEV)',F7.3,14X,'ENERGETIC THRESHOLD(MEV)
     2',F7.3//)
  805 FORMAT(5X,'INITIAL ENERGY(MEV)',F7.2//)
  806 FORMAT(5X,'% ELASTIC SCATTERING ON HYDROGEN',F8.2//)
  807 FORMAT(5X,'% ELASTIC SCATTERING ON CARBON',F8.2//)
  808 FORMAT(5X,'% INELASTIC SCATTERING ON CARBON',F8.2//)
  809 FORMAT(5X,'% ESCAPED NEUTRONS',F8.2//)
  811 FORMAT(5X,'% EFFICIENCY',F8.2/)
  812 FORMAT(5X,'  % ERROR    ',F8.2////)
  900 FORMAT(5F10.0)
  901 FORMAT(5X,'CODE=1 OR 2 IF THE INCIDENCE IS PERP.OR PARAL.',F5.2//)
  902 FORMAT(5X,'-----*-----',20X,'------*------'////)
      GO TO 405
      END
```

```
                 CARG                     DATE = 79150          11/16/27
      SUBROUTINE CARG
C
C-----CARG PROVIDES THE CROSS SECTIONS FOR THE EVENTS CONSODIRED
C     SEM:ARE THE CROSS SECTION FOR THE ELASTIC EVENTS IN THE
C     ENERGERIC RANGE(0.02.0.09)MEV WITH INTERVAL=0.01MEV
C     SE:ARE THE CROSS SECTION FOR ALL THE EVENTS IN THE ENERGETIC
C     RANGE (0.1,15) MEV WITH INTERVAL=0.1).
C     TABM AND TAB ARE THE PROBABILITIES.
C     CARG ALSO PROVIDES THE ANGULAR DISTRIBUTIONS TO CONSIDER THE
C     ANISOTROPY IN THE COLLISIONS N-C AND THE ANGULAR DISTRIBUTION
C     OF THE OUTGOING ALPHA-PARTICLE IN THE REACTION C12(N,ALPHA)BE9.
C     OL(I)  TABULATED DATA OF LIGHT OUTPUT IN THE ENERGY RANGE FROM 20
C     TO 200 KEV
      COMMON/LIL/SEEM(5),SEM(16,2),TABM(16,2)
      COMMON/ANI/ENE,PE(42),AD(21,42),ADN(21)
      COMMON PROB(5),TAB(200,5),SE(200,5)
      COMMON/JM/HN,RHC,A,B,G
       COMMON/ALPHA/ADA(21)
      COMMON/LIGHT/OL(10)
      NCP=5
      NRE=150
      READ(5,13)(OL(I),I=1,10)
      READ(5,6)((SEM(J,I),J=1,8),I=1,2)
      READ(5,4)((SE(J,I),J=1,NRE),I=1,NCP)
      READ(5,7)(PE(I),I=1,42)
      READ(5,8)((AD(J,I),J=1,21),I=1,42)
      READ(5,9)(ADA(I),I=1,21)
      READ(5,14)HN,RHC,A,B,G
      DO 1 J=1,NRE
    1 TAB(J,1)=SE(J,1)/(SE(J,1)+(SE(J,2)+SE(J,3)+SE(J,4)+SE(J,5))/RHC)
      DO 3 I=2,NCP
      DO 2 J=1,NRE
    2 TAB(J,I)=SE(J,I)/(RHC*SE(J,1)+SE(J,2)+SE(J,3)+SE(J,4)+SE(J,5))
    3 CONTINUE
      DO 21 J=1,8
      TABM(J,1)=SEM(J,1)/(SEM(J,1)+SEM(J,2)/RHC)
   21 TABM(J,2)=SEM(J,2)/(RHC*SEM(J,1)+SEM(J,2))
    4 FORMAT(10F8.0)
    6 FORMAT(8F10.0)
    7 FORMAT(16F5.0)
    8 FORMAT(7F10.0)
    9 FORMAT(7F10.0)
   13 FORMAT(10F7.0)
   14 FORMAT(5F10.0)
      RETURN
      END
```

PROBA DATE = 79150 11/16/27

```
      SUBROUTINE PROBA(E)
C-----PROBA CHOOSES THE PROBABILITIES FOR THE NEUTRON ENERGY CONSIDERED
C     AND THESE ONES ARE STORAGED EN THE BLANK COMMON.
      COMMON/LIL/SEEM(5),SEM(16,2),TABM(16,2)
      COMMON PROB(5),TAB(200,5)
      COMMON/JM/HN,RHC,A,B,G
      IF(E-0.1)1,2,2
    2 DO 20 I=1,5
   20 PROB(I)=TAB(IFIX(E*10.+.5001),I)
      GO TO 3
    1 DO 21 I=1,2
   21 PROB(I)=TABM(IFIX((E-0.01)*100.+.5001),I)
      PROB(3)=0
      PROB(4)=0
      PROB(5)=0
    3 CONTINUE
      RETURN
      END
```

```
                              COSENO                DATE = 79150           11/16/27
      SUBROUTINE COSENO(SITA,ALF,BET,GAM)
      COMMON/ANI/ENE,PE(42),AD(21,42),ADN(21)
C
C-----COSENO COMPUTES THE CENTER OF MASS COSINE OF POLAR SCATTERING
C     ANGLE (SITA) IN BOTH CASES: ISOTROPIC AND ANISOTROPIC SCATTERING,
C     AND USES IT TO CALCULATE THE NEW DIRECTION COSINES INTHE LABORA-
C     TORY SYSTEM.
      COMMON/HILDA/RM1,RM2
      COMMON/SEMI/KS
      IF(RM2-12.)10,11,11
   10 R5=RANDOM(KS)
      SITA=(2.*R5)-1.00001
      GO TO 14
   11 R12=RANDOM(KS)
      IF(ENE.GE.10.89)GO TO 16
      DO 1 J=1,41
      D=((PE(J)+PE(J+1)))/2.-PE(J)
      IF(PE(J).LE.(ENE+D))GO TO 15
      IF((ENE+D).LT.PE(1))GO TO 10
   15 IF((ENE+D).LT.PE(J+1))GO TO 12
    1 CONTINUE
   12 L=J
      GO TO 17
   16 L=42
   17 DO 13 I=1,21
   13 ADN(I)=(AD(I,L))/1.67118
      SA=0
      DO 40 K=1,21
      SA=ADN(K)+SA
      IF(R12-SA)21,21,40
   40 CONTINUE
   21 M=K
   22 SITA=FLOAT(M-11)/10.
   14 R8=RANDOM(KS)
      X=R8-0.5
      SSITA=SEGN(X)*SQRT(1.-SITA**2)
    6 R6=RANDOM(KS)
      X=R6
      R7=RANDOM(KS)
      Y=(2.*R7)-1.
      Q=X**2+Y**2
      IF(Q-1.)5,5,6
    5 CPHI=((Y**2)-(X**2))/Q
      SPHI=(2.*X*Y)/Q
      R9=RANDOM(KS)
      X=R9-0.5
      A=SEGN(X)*SQRT(1.+2.*RM2*SITA+(RM2**2))
      CCHI=(1.+RM2*SITA)/A
      SCHI=(RM2*SSITA)/A
      B=1.-GAM**2
      IF(B)7,8,7
    7 R11=RANDOM(KS)
      X=R11-0.5
      BB=SEGN(X)*SQRT(B)
      ALF=ALF*CCHI+(GAM*ALF*SCHI*CPHI-BET*SCHI*SPHI)/BB
      BET=BET*CCHI+(GAM*BET*SCHI*CPHI+ALF*SCHI*SPHI)/BB
      GAM=GAM*CCHI-BB*SCHI*CPHI
      GO TO 9
```

COSENO DATE = 79150 11/16/27

```
8 ALF=SCHI*CPHI
  BET=SCHI*SPHI
  GAM=GAM*CPHI
9 RETURN
  END
```

```
                    HIDR                       DATE = 79150          11/16/27
      SUBROUTINE HIDR(ENE,SITA,P)
C-----HIDR COMPUTES THE ENRGIES OF THE SCATTERD NEUTRON AND THE RECOIL
C     PROTON WHEN THE ELASTIC SCATTERING WITH HYDROGEN TAKES PLACE,
C     AND ALSO COMPUTES THE LIGHT OUTPUT DUE TO THE RECOIL PROTON.
      COMMON/LIGHT/OL(10)
      EPR=ENE*(1.0-SITA)/2.0
      ENE=ENE*(1.0+SITA)/2.0
      IF(EPR-0.02)5,6,6
    5 P=0.00001
      GO TO 4
    6 IF(EPR-0.2)2,2,3
    2 P=OL(IFIX(((EPR+.01)/2.)*100.))
      GO TO 4
    3 T1=.95*EPR
      T2=-.1*EXP(.9*ALOG(EPR))
      T3=-8.*(1.-EXP(T2))
      P=T1+T3
    4 CONTINUE
      RETURN
      END
```

ELAC DATE = 79150 11/16/27

```
C-----ELAC COMPUTES THE SAME AS HIDR IN THE CASE OF ELASTIC SCATTERING
C      WITH CARBON.
       SUBROUTINE ELAC(ENE,SITA,P)
       EC=0.142*ENE*(1.-SITA)
       ENE=ENE-EC
       P=0.01*EC+0.00001
       RETURN
       END
```

```
                  ALPHAN                    DATE = 79150              11/16/27
      SUBROUTINE ALPHAN(P)
C-----ALPHAN COMPUTES  THE ENERGY AND THE LIGHT PULSE OF THE ALPHA
C     PARTICLE AND OF THE BERILIO WHEN THE REACTION C12(N,ALPHA)BE9
C     TAKES PLACES.
C
      COMMON/ALPHA/ADA(21)
      COMMON/ANI/ENE
      R13=RANDOM(KS)
      SB=0
      DO 1 I=1,21
      SB=ADA(I)/127.46+SB
      IF(R13-SB)2,2,1
    1 CONTINUE
    2 M=I
      SITA=FLOAT(M-11)/10.
      Q=5.71
      F=ENE-Q
      A=0.05325*(ENE/F)
      B=0.02367*(ENE/F)
      C=0.28402*(1.-(Q/(12.*F)))
      D=0.63905*(1.-(Q/(12.*F)))
      WRITE(6,4)ENE,A,C,F
    4 FORMAT(5X,'ENE-A-C-F',4F10.4)
      G=2.*SQRT(A*C)
      EAL=F*(B+D+G*SITA)
      EBE=F*(A+C-G*SITA)
      PAL=0.046*EAL+0.007*(EAL**2)
      PBE=.01*EBE
      P=PAL+PBE
      RETURN
      END
```

RANDOM DATE = 79150 11/16/27

```
      FUNCTION RANDOM(KS)
C-----RANDOMCCOMPUTES RANDOM NUMBERS UNIFORMLY DISTRIBUTED BETWEN(0,1)
      IY=KS*65539
      IF(IY)5,6,6
    5 IY=IY+2147483647+1
    6 YFL=IY
      RANDOM=YFL*.4656613E-9
      KS=IY
      RETURN
      END
```

SEGN DATE = 79150 11/16/27

```
      FUNCTION SEGN(X)
C-----SEGN IS ONLY TO PROVIDE A SIGN (+ OR -) IF THE ARGUMENT
C     IS NEGATIVE OR POSITIVE.
      IF(X)10,20,20
   10 SEGN=1.
      RETURN
   20 SEGN=-1.
      RETURN
      END
```

ALAMDA DATE = 79150 11/16/27

```
      FUNCTION ALAMDA(ENE)
C----ALAMDA COMPUTES THE MEAN FREE PATH OF THE NEUTRON
      COMMON/LIL/SEEM(5),SEM(16,2),TABM(16,2)
      COMMON/JM/HN,RHC,A,B,G
      COMMON PROB(5),TAB(200,5),SE(200,5),SEE(5)
      CALL PROBA(ENE)
      IF (ENE-0.09)24,23,23
   23 DO 22 I=1,5
   22 SEE(I)=SE(IFIX(ENE*10.+0.5001),I)
      ALAMDA=1./(HN*(SEE(1)+(SEE(2)+SEE(3)+SEE(4)+SEE(5))/RHC))
      GO TO 25
   24 DO 21 I=1,2
   21 SEEM(I)=SEM(IFIX((ENE-.01)*100.+.5001),I)
      ALAMDA=1./(HN*SEEM(1)+(SEEM(2)/RHC))
   25 CONTINUE
      RETURN
      END
```

O U T F O R S A M P L E P R O B L E M

CODE=1 OR 2 IF THE INCIDENCE IS PERP.OR PARAL. 2.00

SCINTILLATOR RADIUS(CM) 6.000 -HALF-HEIGHT(CM) 3.050

NUMBER OF HISTORIES 40000.

CUT-OFF ENERGY(MEV) 0.020 ENERGETIC THRESHOLD(MEV) 0.900

INITIAL ENERGY(MEV) 2.70

% ELASTIC SCATTERING ON HYDROGEN 45.26

% ELASTIC SCATTERING ON CARBON 25.68

% INELASTIC SCATTERING ON CARBON 0.00

% ESCAPED NEUTRONS 33.05

% EFFICIENCY 30.40

% ERROR 0.91

-----*----- -----*-----

INITIAL ENERGY(MEV) 14.50

% ELASTIC SCATTERING ON HYDROGEN 20.27

% ELASTIC SCATTERING ON CARBON 18.77

% INELASTIC SCATTERING ON CARBON 9.03

% ESCAPED NEUTRONS 56.97

% EFFICIENCY 20.89

% ERROR 1.09

-----*----- -----*-----

Kiadja a Központi Fizikai Kutató Intézet
Felelős kiadó: Szegő Károly
Szakmai lektor: Kardon Béla
Nyelvi lektor: Harvey Shenker
Példányszám: 310 Törzsszám: 79-762
Készült a KFKI sokszorosító üzemében
Budapest, 1979. szeptember hó